TECHNICAL REPORT NO. 2311

N84-27955

EFFECTS OF SPACECRAFT REFLECTIONS ON RF INTERFEROMETER POSITION LOCATION ACCURACY

FINAL REPORT

APRIL 1984

PREPARED UNDER CONTRACT NO. NAS5-27641, TASK IV
FOR GODDARD SPACE FLIGHT CENTER
NATIONAL AERONAUTICS AND SPACE ADMINISTRATION
GREENBELT, MARYLAND 20771

Technical Report Documentation Page

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.

4. Title and Subtitle

Effects of Spacecraft Reflections on RF Interferometer Position Location Accuracy

5. Report Date
April 1984

6. Performing Organization Code

7. Author(s)
Ronald G. Wallace and Fred C. Ward

8. Performing Organization Report No.
TR 2311

9. Performing Organization Name and Address
ORI, Inc.
1400 Spring Street
Silver Spring, MD 20910

10. Work Unit No. (TRAIS)

11. Contract or Grant No.
NAS5-27641, Task IV

12. Sponsoring Agency Name and Address
Instrument Systems Division
NASA/Goddard Space Flight Center
Greenbelt, Maryland 20771

13. Type of Report and Period Covered
Final
March 1983 - April 1984

14. Sponsoring Agency Code
974.0

15. Supplementary Notes
Technical Officer: Jan Turkiewicz

16. Abstract

This report describes one of three study tasks related to the application of an RF interferometer aboard a low-orbiting spacecraft to determine the location of ground-based transmitters. Computer modeling was used to estimate the error in the measured signal angle-of-arrival caused by reflection and diffraction off the spacecraft. Existing computer codes (NEC-BSC) were modified and used to determine the perturbation, due to the spacecraft, in the phase difference between two interferometer antennas, suspended on either side of the spacecraft. This phase perturbation was found as a function of the angle-of-arrival of the signal from a far-field source. The spacecraft antennas were assumed to be circularly polarized with a cardioid pattern. It was found that the perturbation was as much as 13.4° within the $\pm 60^{\circ}$ field-of-view. This suggests that phase calibration and correction of phase measurements are essential for precision position location using this technique.

17. Key Words	18. Distribution Statement
RF Interferometry Antennas Numerical Electromagnetic Codes Satellite-Based Position Location Computer Modeling	

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price
Unclassified	Unclassified	45	

TECHNICAL REPORT NO. 2311

ORI
Silver Spring, Maryland 20910

EFFECTS OF SPACECRAFT REFLECTIONS ON RF INTERFEROMETER POSITION LOCATION ACCURACY

FINAL REPORT

APRIL 1984

PREPARED UNDER CONTRACT NO. NAS5-27641, TASK IV
FOR GODDARD SPACE FLIGHT CENTER
NATIONAL AERONAUTICS AND SPACE ADMINISTRATION
GREENBELT, MARYLAND 20771

TABLE OF CONTENTS

I. INTRODUCTION

The use of RF interferometers has been studied as a technique for radiolocation of surface transmitters from orbit. With an interferometer, one determines a signal's angle of arrival with respect to a baseline by measuring the phase difference between the signals received by antennas at either end of the baseline. If the received signals were exactly proportional to the undisturbed electric field produced by the transmitter that would exist at the ends of the baseline, then the phase difference and the angle of arrival would be exactly related by a simple relation. Unfortunately, the electric fields in the vicinity of the interferometer's antennas are not undisturbed, because the spacecraft and its appendages are constructed of materials that reflect electromagnetic waves. In addition, the antennas themselves disturb the fields. The signals from the antennas are therefore different in amplitude and phase from the signals that would correspond to the undisturbed fields, and the differences are dependent on the angle of arrival. The accuracy of locating emitters with an interferometer may depend to a large extent on how well these phase differences are known as a function of arrival angle. If they are known and are not too great, it should be possible to compensate for them.

The objective of the study reported herein was to estimate the magnitude of the effect of reflections from the spacecraft on interferometer position determination. This was done using a computer program called the Basic Scattering Code (BSC), which is part of a set of programs called the Numerical Electromagnetic Codes (NEC), produced by the Ohio State University

Electroscience Laboratory for the U.S. Navy. This program was provided to ORI, through GSFC, by the National Ocean System Center. The program was modified to accept a particular antenna pattern and to generate the output desired, and was run using representative interferometer/spacecraft configurations. The effect of the reflections from the spacecraft were determined as the difference between the interferometer phase with and without the spacecraft in place.

This report contains the following:

- Specification of the interferometer problem,
- Description of the NEC/BSC program,
- Description of the changes made to the program to solve the interferometer problem, and
- Presentation and interpretation of the results of the program runs.

II. SPECIFICATION OF THE INTERFEROMETER REFLECTION PROBLEM

The geometry of the problem is shown in Figure 2.1. The coordinate system may be related to that of an orbiting spacecraft by the following correspondence:

x-axis: Nadir
y-axis: Spacecraft Velocity Vector
z-axis: Normal to Orbit Plane

The interferometer baseline, of length L, is aligned with the z-axis.

To simplify the problem, we will assume initially that the antennas are circularly polarized in all directions of interest and that the arriving wave is circularly polarized with the same sense. This reduces the problem to a scalar problem.

With the polarization defined, the electric fields E_1, E_2 at the points $z = \pm L/2$ can each be described completely by an amplitude and a phase, as follows:

$E\ (0,\ 0,\ L/2) = A_1 e^{jp_1}$ at antenna 1

$E\ (0,\ 0,\ -L/2) = A_2 e^{jp_2}$ at antenna 2



FIGURE 2.1. GEOMETRY OF THE PROBLEM

The basic interferometer formula relates the difference between the phases of the fields to the direction of signal arrival as follows:

$$p_2 - p_1 = kL \sin \theta$$

where k is the wavenumber, equal to 2π divided by the wavelength.

The electric field at a point may be measured using an antenna. Antennas placed at $z = \pm L/2$ will produce output voltages, V_1, V_2, with amplitudes and phases related to those of the fields by

$$V_{d1} = (B_1 e^{jp'_1})(A_1 e^{jp_1})$$

$$V_{d2} = (B_2 e^{jp'_2})(A_2 e^{jp_2})$$

The parameters B and p' are in general functions of the angle of arrival (θ, ϕ). The phase difference between the voltages from the two antennas is the measurable quantity from which we can infer angle of arrival. It is given by

$$\text{Arg}(V_{d1}) - \text{Arg}(V_{d2}) = p_2 - p_1 + \Delta p_a = kL \sin \theta + \Delta p_a$$

$$\Delta p_a = p'_2(\theta, \phi) - p'_1(\theta, \phi)$$

By carefully matching the antennas, the quantity Δp can, hopefully, be minimized.

Now we introduce an object (the spacecraft) at the origin, midway between the antennas. Reflection and diffraction from the object will produce additional fields, which will generate their own voltages at the antenna outputs. Each antenna's output voltage with the spacecraft in place will be the sum of voltage due to the direct wave, V_d, and the voltage due to the reflected and diffracted wave, V_r. Because the reflected wave field is proportional to the incident wave field, V_r is proportional to V_d, and the total voltage can be written as the product of a complex factor, $Ce^{jp''}$, and the direct wave voltage, as follows:

$$V_1 = (C_1 e^{jp''_1}) V_{d1}$$

$$V_2 = (C_2 e^{jp''_2}) V_{d2}$$

The measurable phases are now the arguments of V_1 and V_2. The difference between these phases is related to the angle of arrival as follows:

$$\text{Arg}(V_2) - \text{Arg}(V_1) = kL\sin\theta + \Delta p_a + \Delta p_r$$

$$\Delta p_r = p''_2(\theta,\phi) - p''_1(\theta,\phi)$$

The quantity $\Delta p_r + \Delta p_a$ can be regarded as the total perturbation of the phase difference due to the antenna and the reflections. If this perturbation is known as a function of angle-of-arrival, it is possible to compensate for the perturbations in the process of finding the position of the emitter from the phase measurements. The compensation may be performed by an iterative procedure which converges on the correct solution. If the perturbations are too large, there may be difficulty in achieving convergence. The magnitude of the reflection-produced phase perturbation, Δp_r, has been estimated using the NEC/BSC computer code.

III. ANALYSIS APPROACH

A methodology has been developed for studying the effects of scattering off a spacecraft body on interferometer angle measurements. A modified version of the Ohio State University Numerical Electromagnetic Code - Basic Scattering Code is used to analyze the user-defined spacecraft and antenna geometry to produce an intermediate file of certain electromagnetic field data. This intermediate file is used by PROGRAM THETA, written by ORI to calculate the effect of scattering on the signal phase difference between the antennas. The error in calculated angle of arrival due to scattering is also determined.

3.1 NUMERICAL ELECTROMAGNETIC CODE - BASIC SCATTERING CODE (BSC)

The Ohio State University Basic Scattering Code was chosen for this project because of its availability and excellent documentation. It is a user-oriented code for the electromagnetic analysis of the radiation from antennas in the presence of complex structures. The code has several applications including the prediction of far zone patterns of antennas in the presence of scattering structures. Simulation of the scattering structures is accomplished by using combinations of perfectly conducting multiple flat plates and finite elliptical cylinders. The code also has a limited finite thin dielectric slab capability. Several options are available for the modeling of antennas, including the capability to model an antenna as a set of half-wavelength dipoles.

The limitations associated with the computer code result mainly from the basic nature of the analysis. The solution is derived from the Geometric Theory of Diffraction which is a high frequency approach. For scattering from plate structures, this means that each plate should have edges at least a wavelength long. In addition, each antenna element should be at least a wavelength from all edges. The use of the Basic Scattering Code restricts the study to considering only perfectly conducting multiple flat plates and finite elliptical cylinders. A further restriction involves the modeling of antennas. They must be represented by sets of apertures or dipoles.

The ideal model for estimating the effect of scattering on a spacecraft-mounted interferometer would consist of receiving antennas mounted on the spacecraft, and a linearly polarized transmitter in the far field with respect to the spacecraft. The model that was actually used in the present study applies the principle of reciprocity to turn the problem around. The antennas mounted on the spacecraft were modeled as transmitters, so that the Basic Scattering Code generated electromagnetic fields at a receiver in the far field. Because of reciprocity, the phase difference that would be measured by the interferometer is equal to the difference in the phases of the fields at the receiver due to each interferometer antenna transmitting by itself. Without additional modification to the Basic Scattering Code, the results are limited to theta and phi receiver polarizations (refer to Figure 2.1). The limitation on the analysis is that the results apply to transmitters that have purely linear polarizations in the theta or phi directions.

A search was made of recent literature to determine if an antenna of the contemplated design (quadrifilar helix) could be approximated by a model consisting of a set of dipoles. No such model was found. The analytical antenna model chosen has a cardioid radiation pattern and is circularly polarized. This model is not part of the Basic Scattering Code and requires a modification to the code discussed below. While not ideal, the cardioid pattern has characteristics that are similar to the pattern of an antenna of quadrifilar helix design. From zero to 65 degrees off axis, the pattern gain

drops about 3 dB. At 90 degrees and beyond where scattered radiation would be expected to enter the antenna, the gain is at least 6 dB less than the on-axis gain.

Basic Scattering Code was installed on the ORI Prime 400 computer system. Some system incompatibilities were found in the original version of the delivered code. With the assistance of the "Installation Hints" section of the Basic Scattering Code User's Manual, minor modifications were made to the code to compile it on the Prime. The installation was verified by running on the Prime a documented example found in the code User's Manual. The Prime version output was compared to the documented example. Some differences in output were noted. These differences were in very small numbers or in least significant digits and can be assumed to be due to the differences in internal precision of the computer installations. The code is considered successfully installed.

The Basic Scattering Code User's Manual documents input command options. Appendix C is a sample BSC input file. An explanation of some of the commands follows.

Command TO: Test options

This command allows the user to set several flags which cause the program to print selected diagnostic data. It allows the user to specify the field components to be included in the results. In this study, the diagnostic flags were set to obtain fields due to the source only, or to obtain all field components.

Command PG: Plate Geometry

This command allows the user to specify plate geometry by the number of corners of the plate and the location of each corner in reference coordinates.

Command PD: Far Zone Pattern

This command enables the user to define the far zone pattern coordinate system. The resulting BSC output is in this pattern coordinate system. The geometry is illustrated in Figure 3.1a. The variables THCX and PHCX define the direction of the x-axis of the pattern coordinate system with respect to the reference coordinates, and THCX and PHCX likewise define the z-axis direction. Note that for the pattern coordinate system and the reference coordinate system to be equivalent, THCX must be 90 degrees. The antenna pattern "cut" is also specified by this command. In this study the great circle pattern cut is used (Figure 3.1b). The parameter PHP is held constant as THP is varied from 0 to 360^{0} to make a pattern cut.

Command SG: Source Geometry

This command enables the user to specify the location and type of source to be used. In this study, sources were modeled as half wavelength dipoles.

Command RT: Rotate-Translate Geometry

This command enables the user to translate and/or rotate the reference coordinate system. This command simplifies the specification of geometry.

Command NP: Next Set of Plates

Command NS: Next Set of Sources

These commands allow the user to void previously input plates and sources and allow specification of a new set of plates and sources.

3.2 MODIFICATIONS MADE TO BASIC SCATTERING CODE

Two modifications were made to the Basic Scattering Code during this study. The first modification was made to the main executive routine and to



FIGURE 3.1a. DEFINITION OF PATTERN COORDINATE SYSTEM



FIGURE 3.1b. GREAT CIRCLE PATTERN CUT

SUBROUTINE OUTPUT. The purpose of this modification was to output certain electromagnetic field data to logical unit 7. The file created by this change is used as the input file for PROGRAM THETA discussed below. The change consisted of adding WRITE statements. Statements added are listed in Appendix A. Note that this modification uses the same logical unit number as the original code had reserved for data to be plotted by the Ohio State University plotting package. The plot option has not been disabled and should not be used with the modified code.

The second modification was a major revision of SUBROUTINE SOURCE, renamed SUBROUTINE SOURCEB. When SOURCEB is used in the Basic Scattering Code, all sources modeled are cardioid. Appendix B is a listing of SUBROUTINE SOURCEB. When SOURCEB is used, the source geometry, relative phase and amplitudes are input into the Basic Scattering code as if the sources were dipoles. (Refer to User's Manual.) Since the internal geometry of the code is not the same as Figure 2.1, some explanation at this point is necessary to understand the analysis of SOURCEB.

Figure 3.3 is a dipole in the Basic Scattering Code coordinate system. SOURCEB uses these same coordinates to define the cardioid source. Note that in Figure 3.3 the dipole's symmetry axis is the z axis. Figure 2.1 uses what is defined in the Basic Scattering Code input file as the pattern coordinate system and in these coordinates the x-axis is nadir. To align the cardioid's symmetry axis with nadir, as required, the internal coordinate system is pitched 90 degrees through the BSC command that specifies the source.

New code generated for SOURCEB was minimal. The source field for a "cardioid antenna" is defined as follows:

$$E_\theta = -A\ (\cos\phi + j\sin\phi)$$
$$E_\phi = A\ (\sin\phi - j\cos\phi)$$
$$\text{where } A = 1.0 + \cos\theta$$

In the code, the source field is calculated and the theta and phi components of the source field are projected onto the x, y, z unit vectors of the internal coordinate system.



FIGURE 3.2. ILLUSTRATION OF ONE-DIMENSIONAL DIPOLE SOURCE

3.3 DIPOLE STUDIES

The first runs made in this study were made with dipole antennas. This was done to simplify the establishment of the methodology. Half wavelength dipole antennas were modeled at positions (5,0,0) and (-5,0,0) in the Basic Scattering code coordinate system. (Units here are meters.) The dipoles were rolled 90 degrees to align them with the y axis of the reference coordinate system. A two meter cube box was modeled, centered at the origin. The geometry of the dipole runs is shown in Figure 3.3. A typical input file for the dipole runs is listed in Appendix C. Recall that the pattern coordinate system is the reference coordinate system, pitched 90 degrees. The pattern coordinate system is given in Figure 2.1.

In each run made of the Basic Scattering Code, the code was executed four times. The first two executions calculated results for each antenna separately, without adding the effects of scattering geometry. The second two executions were for each source separately but added to the direct path (source only) field are the fields scattered off the box. The four sets of data produced by each run can be summarized as follows:

Set	Description
1,3	Source 1
2,4	Source 2
1,2	Source fields only
3,4	Source and Scattered fields

Each set of data contains a record for each pattern position defined in the Basic Scattering Code input file. Each record consists of the following four fields:

Theta and Phi - Define the pattern position
ETHP, EPHP - Phases of Theta and Phi electric field components



FIGURE 3.3. DIPOLE STUDY GEOMETRY, REFERENCE COORDINATE SYSTEM

These four sets of data, along with other values discussed below are output to an intermediate data file for input to PROGRAM THETA.

Note that in addition to being simpler to analyze, since the characteristics of halfwavelength dipole antennas are well known, the dipole studies represent a worst case. This is because the gain of a dipole is the same in the direction of the spacecraft as it is in the direction of nadir. Any improvement in antenna pattern, such as using the "cardioid antenna", should reduce the effects of backscatter into the antenna. The accuracy of calculations of arrival angle should improve. Also the geometry selected is particularly poor for the angle of arrival calculations. Backscatter off a 90 degree wedge enters the antenna pattern at about 76 degrees and corner diffraction enters a little lower. Any improvement in spacecraft geometry which reduces backscatter or enters the antenna pattern at greater angles should improve calculations.

3.4 PROGRAM THETA

The purpose of PROGRAM THETA is to calculate the effects of backscatter on the angle of arrival calculation and the phase difference of the two antennas. A listing of the program is given in Appendix D. An outline of the program follows:

For each set of data read:

- Wavenumber
- Degrees to radians conversion
- Antenna position
- Begin, end and increment for record loop

For each record read:

- Theta
- Phi
- ETHP - Theta component phase
- EPHP - Phi component phase

Calculate distance between the sources

Calculate initial N = integral number of wavelengths between sources

For each record:

Calculate the difference in arrival phases

D(k) where k = 1 - Source only theta component
2 - Source only phi component
3 - Combined theta component
4 - Combined phi component

(where combined is the result of source and backscattered fields)

Calculate the phase perturbation caused by backscatter

DD(1) - Theta component = D(3) - D(1)

DD(2) - Phi component = D(4) - D(2)

For each k (defined above)

Calculate angle of arrival

$T(k) = \arcsin\ ((2\pi N + D(k))\ (WL/\ 2\pi H))$

where WL = wavelength
H = distance between sources

Calculate the differences in angle of arrival calculations between the source only and the perturbed results

Output:

Differences in angle of arrival calculations, theta and phi polarizations

Phase perturbations caused by backscatter, theta and phi polarizations

3.5 CARDIOID ANTENNA PATTERN

The next step in the study was to test the "cardioid antenna" pattern discussed in section 3.2. For this purpose the antenna was defined to be at the origin with no scattering geometry. The results were generally as expected with one exception at exactly theta equals 90 degrees (nadir). Here the antenna gain pattern dropped about 50 dB. No attempt was made to correct

the code for this anomaly. It is mentioned here because it can produce unusual errors in arrival angle calculations for nadir. The "cardioid antenna" pattern is shown in Figure 3.4.

3.6 CARDIOID ANTENNA PATTERN STUDIES

The Basic Scattering Code input files for the cardioid studies are essentially similar to the dipole study input files. The only difference is that the source geometry is not rotated. This orients the axis of the cardioid with nadir.

Two types of runs were performed with cardioids. The difference in runs was in the placement of the scattering geometry. The first set was with the box centered at origin. For this case, we expect to see angle of arrival calculations improve over the dipole sources. The improvement is due to the backscattered fields entering the antenna patterns about 4 dB down from the direct path fields. In the second set of runs the box was displaced one meter such that backscatter would enter the antenna patterns at about the 6 dB point in the antenna gain pattern. Again further improvement in angle of arrival calculations would be expected.



FIGURE 3.4. CARDIOID ANTENNA PATTERN

IV. DISCUSSION OF RESULTS

As described in Section 3, computer runs were made with simple dipole antennas, and then with antennas having a cardioid pattern. This was done to investigate the improvement in performance resulting from the reduction in off-axis antenna gain. The dipole results are presented first, followed by the cardioid results.

4.1 DIPOLE RESULTS

Figure 4.1 shows the amplitude of the electric field due to one dipole antenna located to the right of the box. The field is measured in the plane containing the antenna, the center of the box, and nadir (the xz-plane of Figure 3.3). The direction of the field is perpendicular to that plane (the y-direction). This pattern is the same as that which would describe the output of the antenna when a source was moved through the angle indicated. The scale of the plot is 20 times the logarithm of the electric field intensity, relative to an arbitrary reference level. The oscillations of the pattern are due to the alternate constructive and destructive interference of the direct wave and the secondary waves diffracted from the edges and reflected from the end of the box. The amplitude pattern for a configuration with the antenna located the same distance to the left of the box would be exactly the reverse of the one shown.

Figure 4.2 shows the phase perturbation for a configuration of two dipoles with the box midway between them. This is the phase difference between the antennas as a distant source is rotated through the indicated



FIGURE 4.1. MAGNITUDE OF PHI COMPONENT OF ELECTRIC FIELD FROM DIPOLE ANTENNA



FIGURE 4.2. PHASE PERTURBATION OF PHI COMPONENT OF ELECTRIC FIELD FROM DIPOLE ANTENNAS

angle, relative to the phase difference without the box in place. This reference phase difference is given by (kL)sin θ, where (kL) is 360^0 times the number of wavelengths between the antennas, and θ is the angular coordinate of the plot. The value plotted is thus the difference between the perfect interferometer response and the response with the reflecting box between the antennas. Note that the phase perturbation is perfectly antisymmetrical about zero degrees. This is to be expected considering the amplitudes of the two signals whose phases are being compared, as can be shown as follows:

The amplitude of one of the signals is shown in Figure 4.1, and the amplitude of the other is the mirror image of this. The phase perturbation is a direct function of the ratio of these amplitudes (or their difference, in decibels). Say the (decibel) amplitudes of one of the signals at equal positive and negative angles $\pm\ \theta$ are $A_1(\theta) = x$ and $A_1(-\theta) = y$. The amplitudes of the other signals at these angles are then $A_2(\theta) = y$ and $A_2(-\theta) = x$ because of the mirror image relationship. The amplitude differences (in decibels) are $A_1(\theta) - A_2(\theta) = x-y$ and $A_1(-\theta) - A_2(-\theta) = y-x$, so the amplitude difference is an odd function of θ. The phase difference is therefore also an odd function of θ because the phase difference is directly related to the amplitude difference. This characteristic will hold whenever the configuration of antennas and spacecraft is symmetrical.

Figure 4.2 shows that the phase perturbation oscillates as a function of θ, and the peaks of the oscillations generally increase with the absolute value of θ. This will normally be so with any reasonably-designed antenna layout, because good design would dictate antenna placement resulting in the largest possible unobstructed field-of-view. We also note that the largest phase perturbations are outside of the range of angles from which signals are expected to arrive. The Earth subtends approximately $\pm 60^0$ from nadir at an orbit altitude of 850 km. Within $\pm 60^0$, the maximum perturbation is only 18.9^0, compared with a maximum overall of 73.5^0. Unfortunately, this maximum occurs at an angle of only 38^0 off nadir.

If the phase perturbation caused by reflection is uncompensated, or imperfectly compensated, it will produce an error in the computed angle of arrival. That error is approximated by

$$\Delta\theta = \Delta p/(kL \cos \theta)$$

where Δp is the uncompensated phase perturbation and kL is 2π times the number of wavelengths between the antennas. In the present case, where the wavelength is 75 cm and the antennas are 10 m apart, $kL = 83.8$. The arrival angle error corresponding to the phase perturbation of Figure 4.2, assuming no compensation, is therefore equal to 0.012 times the phase perturbation near nadir, increasing by a factor of $(\cos \theta)^{-1}$ away from nadir. The maximum error within $\theta = \pm 60^{\circ}$ is 0.285, occurring at $\theta = \pm 38^{\circ}$.

4.2 CARDIOID RESULTS

The effect of the reflecting box on the field amplitude with the circularly-polarized "cardioid" antenna is shown in Figure 4.3. The solid line in the figure is the unperturbed field amplitude (in decibels), due to the antenna alone. This is identical to the pattern in Figure 3.4. The dotted line is the magnitude of the total field, considering both polarizations, when the box is placed to the left of the antenna. The total field amplitude is defined as the sum of the squares of the amplitudes of θ-directed field and of the ϕ-directed field. It is proportional to the power flux density. The variation in the amplitude with θ is seen to be very similar to that for the dipole case given in Figure 4.1. The amplitude of the ϕ-directed field alone is nearly identical to that of Figure 4.1, except for a general reduction of the amplitude with off-nadir angle, due to the cardioid pattern.

The phase perturbation with cardioid antennas is shown in Figure 4.4. Comparing this with Figure 4.2 shows that the two curves are nearly identical. The general shape of the curves is the same, but the amplitude of the cardioid curve is less than the dipole curve. The difference between them at the peaks varies from 29 to 56 percent, averaging 39 percent.

FIGURE 4.3. MAGNITUDE OF TOTAL ELECTRIC FIELD FROM CARDIOID ANTENNA





FIGURE 4.4. PHASE PERTURBATION OF PHI COMPONENT OF ELECTRIC FIELD FROM CARDIOID ANTENNAS

The maximum phase perturbation within $\pm 60^{\circ}$ for the cardioid antennas is 13.4°, compared with 18.9° for the dipoles. The maximum angle of arrival perturbation within $\pm 60^{\circ}$ for the cardioids is 0.20°, occurring at $\pm 38^{\circ}$.

The curves presented up to now have all applied to the ϕ-directed electric field, measured in the $\phi_p = 0$ plane, containing the antenna baseline and the nadir direction. Computer runs were also made for the θ-directed field and for other planes. Figure 4.5 is the phase perturbation for the component of electric field in the θ-direction. This applies to ground transmitters having linear polarization oriented parallel to the baseline. We note that the phase perturbation is comparable to that of the ϕ-directed polarization for the $\pm 60^{\circ}$ range of θ. The assymetry of the curve and the large phase excursions at large negative values of θ cannot be readily explained, but they may be related to depolarization of the wave when it is diffracted by the edges of the box. Figure 4.6 shows the phase perturbations for the ϕ-directed field measured on the $\phi_p = 30^{\circ}$ plane. They are smaller than in the case shown for $\phi = 0$, but like the previous case, they show large excursions between $\theta = -70^{\circ}$ and $\theta = -80^{\circ}$ and are not symmetrical. These last two plots demonstrate that the special case of ϕ polarization and the $\phi = 0$ plane used for most of the computer runs does not give atypical results.

To see the effect of altering the antenna/spacecraft configuration, a run was made with the box displaced up by half the length of its side. This displacement results in the lower side of the box containing the antenna baseline, as suggested in the sketch in Figure 4.7. This figure shows the phase perturbation for the displaced box case. Because the box is being moved further out of the field-of-view of the antennas, one would expect a reduction in the phase perturbation. Comparison of Figure 4.7 with Figure 4.2 shows that this happens as expected. The excursions of the phase perturbations are generally reduced, but they are increased somewhat in the range of θ between -20° and $+20^{\circ}$. Also, the angles at which the largest perturbations occur move away from nadir ($\theta = 0$).



FIGURE 4.5. PHASE PERTURBATION OF THETA COMPONENT OF ELECTRIC FIELD FROM CARDIOID ANTENNAS

FIGURE 4.6. PHASE PERTURBATION OF PHI COMPONENT OF ELECTRIC FIELD FOR CARDIOID ANTENNAS, WITH MEASUREMENT PLANE INCLINED 30°



FIGURE 4.7. PHASE PERTURBATION OF PHI COMPONENT OF ELECTRIC FIELD FOR CARDIOID ANTENNAS, WITH BOX DISPLACED ONE METER



APPENDIX A
MODIFICATIONS TO BSC MAIN EXECUTIVE AND SUBROUTINE OUTPUT

1. In the BSC main executive the following insertions were made following line 2199

```
WRITE(7,619) WK, DPR, ((XSS(I, J), J = 1, 3), I = 1, 2)

619 FORMAT (8F 11.6)
```

2. In the BSC subroutine output the following insertions were made:

following line 51 -

```
WRITE(7,505) NBN, NEN, NSN

505 FORMAT (3I3)
```

following line 99 -

```
WRITE(7,504) THI, PHI, ETHP, EPHP

504 FORMAT(4F 11.6)
```

APPENDIX B

SOURCE LISTING FOR SUBROUTINE SOURCE B

```
      SUBROUTINE SOURCE(ES,VI,VAX)
C!!!
C!!!  TRANSMITTER'S SOURCE FIELD PATTERN FACTOR IS COMPUTED.
C!!!
      COMPLEX ES(3),EH,FA,WS,CJ,CPI4
      COMPLEX EFED(1),HFED(1),EFD,HFD,ET,EP,A
      DIMENSION VAX(3,3),VI(3),XA(3)
      COMMON/SORDAT/IM,H,HAW,FACTOR
      COMMON/SORARY/WS(30),XSS(30,3),MSA(2,30),MS,MSP,MSPP
      COMMON/SORINF/XS(3),VXS(3,3)
      COMMON/IMAINF/XI(14,14,3),VXI(3,3,14)
      COMMON/GEOMEL/AC(5),BC(5),ZC(2,5),TCR(2,5),XCL(3,5)
     2,VCL(3,3,5),MCX
      COMMON/IMCINF/XIC(2,3,5),VXIC(3,3,10)
      COMMON/PIS/PI,TPI,DPR,RPD
      COMMON/COMP/CJ,CPI4
      COMMON/WAVE/WK,WL
      COMMON/FEDDAT/EFED,HFED
      COMMON/LIMIT/SML,SMLR,SMLT,BIG
C!!!  TAKE DOT PRODUCTS OF THE RADIATION DIRECTION UNIT
C!!!  VECTOR AND SOURCE COORDINATE SYSTEM (PRIMED) AXES
C!!!  UNIT VECTORS TO OBTAIN THP AND PHP (PROPAGATION
C!!!  ANGLES IN THE SOURCE COORD SYSTEM)
      CTHP=VAX(3,1)*VI(1)+VAX(3,2)*VI(2)+VAX(3,3)*VI(3)
      RDX=VAX(1,1)*VI(1)+VAX(1,2)*VI(2)+VAX(1,3)*VI(3)
      RDY=VAX(2,1)*VI(1)+VAX(2,2)*VI(2)+VAX(2,3)*VI(3)
      STHP=SQRT(RDX*RDX+RDY*RDY)
      IF(STHP.LT.SMLR) STHP=SMLR
      CPHP=RDX/STHP
      SPHP=RDY/STHP
C!!!  CALCULATE THETA OR PHI POLARIZATION UNIT VECTOR FOR RAY
C!!!  IN SOURCE COORD SYS AND REPRESENT WITH X,Y,Z
C!!!  COMPONENTS IN THE REFERENCE COORDINATE SYSTEM
      XTH=VAX(1,1)*CPHP*CTHP+VAX(2,1)*SPHP*CTHP-VAX(3,1)*STHP
      YTH=VAX(1,2)*CPHP*CTHP+VAX(2,2)*SPHP*CTHP-VAX(3,2)*STHP
      ZTH=VAX(1,3)*CPHP*CTHP+VAX(2,3)*SPHP*CTHP-VAX(3,3)*STHP
      XPH=-VAX(1,1)*SPHP+VAX(2,1)*CPHP
      YPH=-VAX(1,2)*SPHP+VAX(2,2)*CPHP
```

```
      ZPH=-VAX(1,3)*SPHP+VAX(2,3)*CPHP
      A = CMPLX(CTHP+1.0,0.0)
      ET = -A * CMPLX(CPHP,-SPHP)
      EP =  A * CMPLX(SPHP, CPHP)
      ES(1) = XTH * ET + XPH * EP
      ES(2) = YTH * ET + YPH * EP
      ES(3) = ZTH * ET + ZPH * EP
      RETURN
      END
```

APPENDIX C

SAMPLE INPUT TO BSC FOR DIPOLE STUDY

```
CE: DIPOLE TEST W/ 2M BOX AT ORIGIN, FOUR EXECUTIONS, RUN 6
TO: CALCULATE SOURCE FIELD ONLY
0,0,0,1
0,0,0
1
0,0,0,0,0,0
0,0,0,0,0
0,0,0,0
UN: GEOMETRY UNITS IN METERS
1
US: SOURCE UNITS IN WAVELENGTHS
0
FR: FREQUENCY (GIGAHERTZ)
0.400
SG: SOURCE GEOMETRY
5.0,0.,0.
90.,90.,0.,0.
-1,0.5,0.
1.,0.
PD: FAR ZONE PATTERN
-90,0,0,0
0,0
0,180,1
RD: FAR ZONE RANGE
10000
LP: LINE PRINTER OUTPUT
1
XQ: EXECUTE CODE
NS: NEXT SET OF SOURCES
SG:  SECOND SOURCE
-5.0,0.,0.
90.,90.,0.,0.
-1,0.5,0.
1.,180.
XQ: EXECUTE CODE
NS: BACK TO FIRST SOURCE
SG: FIRST SOURCE
5.0,0.,0.
```

```
90.,90.,0.,0.
-1,0.5,0.
1.,0.
PG: PLATE GEOMETRY, 2 METER BOX
4,0
1.,-1.,1.
1.,1.,1.
-1.,1.,1.
-1.,-1.,1.
PG:
4,0
-1.,-1.,-1.
-1.,1.,-1.
1.,1.,-1.
1.,-1.,-1.
PG:
4,0
1.,1.,1.
1.,1.,-1.
-1.,1.,-1.
-1.,1.,1.
PG:

4,0
-1.,-1.,-1.
1.,-1.,-1.
1.,-1.,1.
-1.,-1.,1.
PG:
4,0
1.,1.,1.
1.,-1.,1.
1.,-1.,-1.
1.,1.,-1.
PG:
4,0
-1.,-1.,-1.
-1.,-1.,1.
```

```
-1.,1.,1.
-1.,1.,-1.
TO: CALCULATE ALL FIELDS
0,0,0,1
1,1,0
1
1,1,1,1,1,1
1,1,1,1,1
1,1,1,1
XQ: EXECUTE CODE
NS: BACK TO SECOND SOURCE
SG: SECOND SOURCE
-5.,0.,0.
90.,90.,0.,0.
-1,0.5,0.
1.,180
XQ: EXECUTE CODE
EN: END PROGRAM
```

APPENDIX D

SOURCE LISTING FOR PROGRAM THETA

```
C     PROGRAM THETA
C  CALCULATES IDEAL INTERFEROMETER ANGLE MEASUREMENT AND COMPARES WITH ANGLE
C     MEASUREMENT PERTURBED BY GEOMETRY
C
C
      DIMENSION THI(181), PHI(181), ETHP(4,181), EPHP(4,181), XSS(4,3)
      DIMENSION D(4), T(6), DD(2)
C
C  FORMATS
  101 FORMAT(3I3)
  102 FORMAT(4F11.6)
  103 FORMAT(1H ,'DISTANCE BETWEEN SOURCES = ',F11.6)
  104 FORMAT(1H ,'WAVENUMBER = ',F11.6)
  105 FORMAT(28X,'SOURCE ONLY',14X,'COMBINED',17X,'DELTA')
  106 FORMAT(3(5X,'THETA',6X,'PHI',5X)/)
  107 FORMAT(1H )
  109 FORMAT(29X,'ANGLE OF ARRIVAL',8X,'PHASE',/,29X,'PERTURBATION',
     C12X,'PERTURBATION')
  110 FORMAT(8F11.6)
  111 FORMAT(8F12.6)
      WRITE(6,107)
C
C  LOOP THROUGH EACH SET OF DATA
C     I = 1,3 SOURCE 1
C     I = 2,4 SOURCE 2
C     I = 1,2 SOURCE ONLY
C     I = 3,4 COMBINED
C
      DO 10 I = 1, 4
C
C     READ WAVE NUMBER, DEGREES TO RADIANS, SOURCE LOCATIONS
        READ(5,110) WK, DPR, (XSS(I,J),J=1,3)
C
C     READ BEGIN, END, INCREMENT
        READ (5,101) NBN, NEN, NSN
C
C     READ EACH SET OF PHASE DATA
        J = 0
```

```
        DO 20 J1 = NBN, NEN, NSN
          J = J + 1
          READ (5,102) THI(J), PHI(J), ETHP(I,J), EPHP(I,J)
 20     CONTINUE
 10   CONTINUE
C
C  CALCULATE H - DISTANCE BETWEEN THE TWO SOURCES
      H = 0
      DO 30 J = 1, 3
        X = XSS(1,J) - XSS(2,J)
        H = H + X*X
 30   CONTINUE
      H = SQRT(H)
C
C  WRITE DISTANCE BETWEEN SOURCES, WAVE NUMBER AND HEADER
      WRITE(6,103) H
      WRITE(6,104) WK
      WRITE(6,107)
      WRITE(6,109)
      WRITE(6,106)
C
C  CALCULATE CONSTANT
      C = 1. / (WK * H)
C
C  CALCULATE INITIAL N
      WL = 6.283185307 / WK
      N = -INT(H/WL)
      XN = FLOAT(N)
      X = -360.0
C
C  CALCULATE ANGLES
C
C  LOOP THROUGH DATA
      J = 0
      DO 40 J1 = NBN, NEN, NSN
        J = J + 1
        D(1) = ETHP(1,J) - ETHP(2,J)
        D(2) = EPHP(1,J) - EPHP(2,J)
```

```
        D(3) = ETHP(3,J) - ETHP(4,J)
        D(4) = EPHP(3,J) - EPHP(4,J)
C
C       CALCULATE PHASE PERTURBATIONS
        DD(1) = D(1) - D(3)
        DD(2) = D(2) - D(4)
C
C       LOOP THROUGH K
C         K = 1 SOURCE ONLY THETA
C         K = 2 SOURCE ONLY PHI
C         K = 3 COMBINED THETA
C         K = 4 COMBINED PHI
C
C       TEST FOR NEXT N
            IF(X.GT.D(2)) XN = XN + 1.
            X = D(2)
C
          DO 50 K = 1, 4
C           CALCULATE THETA
              A = ((2.0 * XN * 3.141592654) + (D(K) / DPR)) * C
              T(K) = ASINF(A) * DPR + 90.0
 50       CONTINUE
C
C     OUTPUT RESULTS
        T(5) = T(1) - T(3)
        T(6) = T(2) - T(4)
        WRITE(6,111) THI(J),PHI(J),T(5),T(6),DD(1),DD(2)
C       WRITE(6,110) THI(J),  PHI(J),  (T(K),K=1,6)
C       WRITE(6,110) THI(J),PHI(J),XN,X,(D(K),K=1,4)
C       WRITE(6,110) (ETHP(K,J),K=1,4),(EPHP(K,J),K=1,4)
C       WRITE(6,107)
C
 40     CONTINUE
C
        STOP
        END
C
        FUNCTION ASINF(X)
```

```
C       PROGRAM THETA


        T = 1 - X * X
        IF (T.LT.0.) T = 0.
        ASINF = ATAN2(X,SQRT(T))
        RETURN
        END
```